



UNION INTERNATIONALE DES TÉLÉCOMMUNICATIONS

CCITT

Z.100 Annexe C1

COMITÉ CONSULTATIF
INTERNATIONAL
TÉLÉGRAPHIQUE ET TÉLÉPHONIQUE

(11/1988)

SÉRIE Z: LANGAGES ET ASPECTS INFORMATIQUES
GÉNÉRAUX DES SYSTÈMES DE
TÉLÉCOMMUNICATION

Langage de spécification et de description fonctionnelles
(LDS)

Critères d'utilisation des techniques de description
formelles (TDF)

**RÉSUMÉ DE LA SYNTAXE GRAPHIQUE
CONCRÈTE**

Réédition de la Recommandation du CCITT Z.100
Annexe C1, publiée dans le Livre Bleu, Fascicule X.1
(1988)

NOTES

1 La Recommandation Z.100 Annexe C1 du CCITT a été publiée dans le fascicule X.1 du Livre Bleu. Ce fichier est un extrait du Livre Bleu. La présentation peut en être légèrement différente, mais le contenu est identique à celui du Livre Bleu et les conditions en matière de droits d'auteur restent inchangées (voir plus loin).

2 Dans la présente Recommandation, le terme «Administration» désigne indifféremment une administration de télécommunication ou une exploitation reconnue.

Résumé de la syntaxe graphique concrète**C1.1 Introduction****C1.1.1 Métalangage**

En ce qui concerne la grammaire graphique, le métalangage décrit au § 1.5.2 de la Recommandation relative au LDS est complété par les métasymboles suivants:

- a) **contains**
- b) **is associated with**
- c) **is followed by**
- d) **is connected to**
- e) **set**

Le métasymbole **set** est un opérateur postfixé applicable aux éléments syntaxiques entourés par des accolades qui le précèdent immédiatement et désignant un ensemble (non classé) d'éléments. Il peut s'agir, notamment, de n'importe quel élément syntaxique, qui doit alors figurer avant le métasymbole **set**. Par exemple:

`{ { <zone texte de système> } * { <diagramme de macro> } * <zone interaction de blocs> } set`


est un ensemble de <zones texte de système> dont le nombre est égal ou supérieur à zéro, <de diagrammes de macro> dont le nombre est égal ou supérieur à zéro et d'une <zone interaction de blocs>.

Tous les autres métasymboles sont des opérateurs infixes ayant un symbole graphique non-terminal comme argument de gauche. L'argument de droite est soit un groupe d'éléments syntaxiques entouré d'accolades soit un élément syntaxique unique. Si le membre de droite d'une règle de production comporte un symbole graphique non-terminal comme premier élément et contient un ou plusieurs de ces opérateurs infixes, le symbole graphique non-terminal est alors l'argument de gauche de chacun de ces opérateurs infixes. Un symbole graphique non-terminal est un non-terminal dont le mot «symbole» est placé avant le signe >.


Le métasymbole **contains** indique que son argument de droite doit être placé à l'intérieur de son argument de gauche et, le cas échéant, à l'intérieur du <symbole d'extension de texte>. Exemple:

`<référence de bloc> ::=`
`<symbole de bloc> contains <nom de bloc>`

`<symbole de bloc> ::=`



signifie ce qui suit:



Le métasymbole **is associated with** indique que son argument de droite est associé, sur le plan logique, à son argument de gauche (comme s'il était «contenu» dans cet argument, l'association précise étant définie par les règles appropriées applicables aux dessins).

Le métasymbole **is followed by** signifie que son argument de droite suit (tant sur le plan logique que dans le dessin) son argument de gauche.

Le métasymbole **is connected to** signifie que son argument de droite est relié (tant sur le plan logique que dans le dessin) à son argument de gauche.

C1.1.2 Règles générales

C1.1.2.1 Subdivision des diagrammes

La définition ci-après de la subdivision des diagrammes ne fait pas partie de la grammaire graphique concrète mais le même métalangage est employé.

<page> ::=
 <symbole de cadre> **contains**
 <zone d'en-tête> <zone de numéro de page>
 { <unité syntaxique> }*

<zone d'en-tête> ::=
 <symbole de texte implicite> **contains** <en-tête>

<zone de numéro de page> ::=
 <symbole de texte implicite> **contains** [<numéro de page> [(<nombre de pages>)]]

<numéro de page> ::=
 <nom de littéral>

<nombre de pages> ::=
 <nom de littéral naturel>


<page> étant un non-terminal de départ, il n'est donc mentionné dans aucune règle de production. Un diagramme peut être subdivisé en un certain nombre de <page>, auquel cas le <symbole de cadre> délimitant le diagramme ainsi que l'<en-tête> du diagramme sont remplacés par un <symbole de cadre> et un <en-tête> pour chaque page.

Le <symbole de texte implicite> n'est pas représenté mais il est indiqué implicitement afin qu'une distinction claire soit établie entre la <zone d'en-tête> et la <zone de numéro de page>. La <zone d'en-tête> est placée au coin supérieur gauche du <symbole de cadre>. La <zone de numéro de page> est placée au coin supérieur droit du <symbole de cadre>. L'<en-tête> et l'<unité syntaxique> dépendent du type de diagramme.

C1.1.2.2 Commentaire

<zone de commentaire> ::=
 <symbole de commentaire> **contains** <texte>
 is connected to <symbole d'association pointillé>

<symbole de commentaire> ::=



<symbole d'association pointillé> ::=

Une extrémité du <symbole d'association pointillé> doit être reliée au milieu du segment vertical du <symbole de commentaire>.

Un <symbole de commentaire> est relié à n'importe quel symbole graphique par l'intermédiaire d'un <symbole d'association pointillé>. Le <symbole de commentaire> est considéré comme un symbole clos lorsque le rectangle est complété (de façon imaginaire). Il contient un texte de commentaire en rapport avec le symbole graphique.

C1.1.2.3 Extension de texte

<zone d'extension de texte> ::=
 <symbole d'extension de texte> **contains** <texte>
 is connected to <symbole d'association continu>

<symbole d'extension de texte> ::
 <symbole de commentaire>

<symbole d'association continu> ::=

Un <symbole d'extension de texte> est relié à n'importe quel symbole graphique par l'intermédiaire d'un <symbole d'association continu>. Le <symbole d'extension de texte> est considéré comme un symbole clos lorsque le rectangle est complété (de façon imaginaire).

Une extrémité du <symbole d'association continu> doit être reliée au milieu du segment vertical du <symbole d'extension de texte>.

Le texte contenu dans le <symbole d'extension de texte> fait suite au texte inscrit à l'intérieur du symbole graphique et il est considéré comme faisant partie de ce symbole.

C1.2 Définition de système

<définition de système concret> ::=
 { <définition de système> | <diagramme de système> } { <définition différée> }*

<définition différée> ::=
 <définition>
 | <diagramme>

<diagramme> ::=
 <diagramme de bloc>
 | <diagramme de processus>
 | <diagramme de procédure>
 | <diagramme de sous-structure de bloc>
 | <diagramme de sous-structure de canal>
 | <diagramme de service>
 | <diagramme de macro>

C1.3 Diagramme de système

<diagramme de système> ::=
 <symbole de cadre> **contains**
 { <en-tête de système>
 { { <zone texte de système> }*
 { <diagramme de macro> }*
 <zone interaction de blocs> }set }

<symbole de cadre> ::=



<en-tête de système> ::=
 SYSTEM <nom de système>

<zone texte de système> ::=
 <symbole de texte> **contains**
 { <définition de signal>
 | <définition de liste de signaux>
 | <définition de données>
 | <définition de macro>
 | <définition de sélection> }*

<symbole de texte> ::=



```


<zone interaction de blocs> ::=
    { <zone de bloc> | <zone de définition de canal> }+

<zone de bloc> ::=
    <référence graphique de bloc>
    | <diagramme de bloc>

<référence graphique de bloc> ::=
    <symbole de bloc> contains <nom de bloc>

<symbole de bloc> ::=

```



```

<zone de définition de canal> ::=
    <symbole de canal>
    is associated with <nom de canal>
        { [ { <identificateur de canal> | <identificateur de bloc> } ]
          <zone de liste de signaux>
          [ <zone de liste de signaux> ] set }
    is connected to { <zone de bloc>
        { <zone de bloc> | <symbole de cadre> }
        [ <zone association de sous-structure de canal> ] } set

```

L'<identificateur de canal> identifie une voie externe reliée au <diagramme de sous-structure de bloc> délimité par le <symbole de cadre>. L'<identificateur de bloc> identifie un bloc externe comme étant un point d'extrémité de canal du <diagramme de sous-structure de canal> délimité par le <symbole de cadre>.

```

<symbole de canal> ::=
    <symbole de canal 1>
    | <symbole de canal 2>
    | <symbole de canal 3>

```

```

<symbole de canal 1> ::=


```



```

<symbole de canal 2> ::=

```



```

<symbole de canal 3> ::=

```



```

<zone de liste de signaux> ::=
    <symbole de liste de signaux> contains <liste de signaux>

```

```

<symbole de liste de signaux> ::=

```



C1.4 Diagramme de bloc

```

<diagramme de bloc> ::=
    <symbole de cadre>
    contains { <en-tête de bloc>
        { { <zone texte de bloc> }*
          { <diagramme de macro> }*
          [ <zone d'interaction de processus> ]
          [ <zone de sous-structure de bloc> ] } set }
    is associated with { <identificateur de canal> }*

```

L' <identificateur de canal> identifie un canal relié à un acheminement de signal dans le <diagramme de bloc>. Il est placé à l'extérieur du <symbole de cadre> près du point d'extrémité de l'acheminement de signal au <symbole de cadre>. Si le <diagramme de bloc> ne contient pas une <zone d'interaction de processus>, il doit alors contenir une <zone de sous-structure de bloc>.

<en-tête de bloc> ::=
 BLOCK { <nom de bloc> | <identificateur de bloc> }

<zone texte de bloc> ::=
 <zone texte de système>

<zone d'interaction de processus> ::=
 { <zone de processus>
 | <zone de ligne de création>
 | <zone de définition d'acheminement de signal> }+

<zone de processus> ::=
 <référence graphique de processus>
 | <diagramme de processus>

<référence graphique de processus> ::=
 <symbole de processus> **contains** { <nom de processus>[<nombre d'instances>]}

<symbole de processus> ::=



<zone de ligne de création> ::=
 <symbole de ligne de création>
is connected to { <zone de processus> <zone de processus> }

<symbole de ligne de création> ::=
 ----->

<zone de définition d'acheminement de signal> ::=
 <symbole d'acheminement de signal>
is associated with { <nom d'acheminement de signal>
 { [<identificateur de canal>]
 <zone de liste de signaux>
 [<zone de liste de signaux>]set }
is connected to
 { <zone de processus> { <zone de processus> | <symbole de cadre > }set }

Lorsque le <symbole d'acheminement de signal> est relié au <symbole de cadre>, l' <identificateur de canal> identifie alors un canal auquel l'acheminement de signal est relié.

<symbole d'acheminement de signal> ::=
 <symbole d'acheminement de signal 1>
 | <symbole d'acheminement de signal 2>

<symbole d'acheminement de signal 1> ::=
 ----->

<symbole d'acheminement de signal 2> ::=
 <----->

C1.5 Diagramme de processus

<diagramme de processus> ::=
 <symbole de cadre>
contains { <en-tête de processus>
 { { <zone texte de processus> }*
 { <zone de procédure> }*
 { <diagramme de macro> }*
 { <zone de graphe de processus> | <zone d'interaction de service> }set }
[is associated with { <identificateur d'acheminement de signal> }+]


L' <identificateur d'acheminement de signal> identifie un acheminement de signal externe associé à un acheminement de signal dans le <diagramme de processus>. Il est placé à l'extérieur du <symbole de cadre> près du point d'extrémité de l'acheminement de signal au <symbole de cadre>.

<en-tête de processus> ::=
 PROCESS { <nom de processus> | <identificateur de processus> }
 [<nombre d'instances> <fin>] [<paramètres formels>]

<zone texte de processus> ::=
 <symbole de texte> **contains** [<ensemble de signaux d'entrée valables>]
 { <définition de signal>
 | <définition de liste de signaux>
 | <définition de variable>
 | <définition de vue>
 | <définition d'import>
 | <définition de données>
 | <définition de macro>
 | <définition de temporisateur>
 | <définition de sélection> }*

<référence graphique de procédure> ::=
 <symbole de procédure> **contains** <nom de procédure>

<symbole de procédure> ::=



<zone de graphe de processus> ::=
 <zone de départ> { <zone d'état> | <zone de connecteur d'entrée> }*

<zone de départ> ::=
 <symbole de départ> **is followed by** <zone de transition>

<symbole de départ> ::=



<zone de transition> ::=
 [<zone de chaîne de transition>] **is followed by**
 { <zone d'état>
 | <zone état suivant>
 | <zone de décision>
 | <symbole d'arrêt>
 | <zone de fusion>
 | <zone de connecteur de sortie>
 | <symbole de retour>
 | <zone d'option de transition> }

<zone de fusion> ::=
 <symbole de fusion> **is connected to** <symbole de ligne de flot>

<symbole de fusion> ::=
 <symbole de ligne de flot>

<symbole de ligne de flot> ::=



<zone de chaîne de transition> ::=
 { <zone de tâche>
 | <zone de sortie>
 | <zone de sortie prioritaire>
 | <zone d'initialisation>
 | <zone de réinitialisation>
 | <zone d'export>
 | <zone de demande de création>
 | <zone d'appel de procédure> }
 [is followed by <zone de chaîne de transition>]


<zone de tâche> ::=
 <symbole de tâche> **contains** <corps de tâche>

<symbole de tâche> ::=




<zone de sortie> ::=
 <symbole de sortie> **contains** <corps de sortie>

<symbole de sortie> ::=



<zone de demande de création> ::=
 <symbole de demande de création> **contains** <corps de création>

<symbole de demande de création> ::=



<zone d'appel de procédure> ::=
 <symbole d'appel de procédure> **contains** <corps d'appel de procédure>

<symbole d'appel de procédure> ::=



<zone d'état> ::=
 <symbole d'état> **contains** <liste d'état> **is associated with**
 { <zone d'association d'entrée>
 | <zone d'association d'entrée prioritaire>
 | <zone d'association de signal continu>
 | <zone d'association de mise en réserve> }*

<symbole d'état> ::=



<zone d'association d'entrée> ::=
 <symbole d'association continu> **is connected to** <zone d'entrée>

<zone d'entrée> ::=
 <symbole d'entrée> **contains** <liste d'entrée>
 is followed by [<zone de condition de validation>] <zone de transition > }

<symbole d'entrée> ::=



<zone d'association de mise en réserve> ::=
 <symbole d'association continu> **is connected to** <zone de mise en réserve>

<zone de mise en réserve> ::=
 <symbole de mise en réserve> **contains** <liste de mise en réserve>

<symbole de mise en réserve> ::=



<zone de connecteur d'entrée> ::=
 <symbole de connecteur d'entrée> **contains** <nom de connecteur>
 is followed by <zone de transition>


<symbole de connecteur d'entrée> ::=



<zone état suivant> ::=
 <symbole d'état> **contains** <corps état suivant>

<zone de décision> ::=
 <symbole de décision> **contains** <question>
 is followed by
 { { <partie de réponse graphique> <partie graphique autre> } set
 | { <partie de réponse graphique> { <partie de réponse graphique> } + [<partie graphique
 autre>] set }

<symbole de décision> ::=



<réponse graphique> ::=
 <réponse> | (<réponse>)

<partie de réponse graphique> ::=
 <symbole de ligne de flot> **is associated with** <réponse graphique>
 is followed by <zone de transition>

<partie graphique autre> ::=
 <symbole de ligne de flot> **is associated with** ELSE
 is followed by <zone de transition>

<zone d'initialisation> ::=
 <symbole de tâche> **contains** <initialisation>

<zone de réinitialisation> ::=
 <symbole de tâche> **contains** <réinitialisation>

<symbole d'arrêt> ::=



<zone de connecteur de sortie> ::=
 <symbole de connecteur de sortie> **contains** <nom de connecteur>

<symbole de connecteur de sortie> ::=
 <symbole de connecteur d'entrée>

C1.6 *Diagramme de procédure*

<diagramme de procédure> ::=
 <symbole de cadre> **contains** { <en-tête de procédure>
 { { <zone texte de procédure> }*
 { <zone de procédure> }*
 { <diagramme de macro> }*
 <zone de graphe de procédure> }set }

<en-tête de procédure> ::=
 PROCEDURE { <nom de procédure> | <identificateur de procédure> }
 [<paramètres formels de procédure>]


<zone de procédure> ::=
 <référence graphique de procédure>
 | <diagramme de procédure>

<zone texte de procédure> ::=
 <symbole de texte> **contains**
 { <définition de variable>
 | <définition de données>
 | <définition de macro>
 | <définition de sélection> }*

<zone de graphe de procédure> ::=
 <zone de départ de procédure> { <zone d'état> | <zone de connecteur d'entrée> }*

<zone de départ de procédure> ::=
 <symbole de départ de procédure> **is followed by** <zone de transition>

<symbole de départ de procédure> ::=



<symbole de retour> ::=



C1.7 *Sous-structure de bloc*

<zone de sous-structure de bloc> ::=
 <référence graphique de sous-structure de bloc>
 | <diagramme de sous-structure de bloc>

<référence graphique de sous-structure de bloc> ::=
 <symbole de sous-structure de bloc> **contains** <nom de sous-structure de bloc>

<symbole de sous-structure de bloc> ::=
 <symbole de bloc>

<diagramme de sous-structure de bloc> ::=
 <symbole de cadre>
 contains { <en-tête de sous-structure de bloc>
 { { <zone texte de sous-structure de bloc> }*
 { <diagramme de macro> }*
 <zone interaction de blocs> }set }
 is associated with { <identificateur de canal> }*

L'<identificateur de canal> identifie un canal relié à un sous-canal dans le <diagramme de sous-structure de bloc>. Il est placé à l'extérieur du <symbole de cadre> près du point d'extrémité du sous-canal au <symbole de cadre>.

<en-tête de sous-structure de bloc> ::=
 SUBSTRUCTURE { <nom de sous-structure de bloc> | <identificateur de sous-structure de bloc> }

<zone texte de sous-structure de bloc> ::=
 <zone texte de système>

C1.8 *Sous-structure de canal*

<zone association de sous-structure de canal> ::=
 <symbole d'association pointillé>
 is connected to <zone de sous-structure de canal>

<zone de sous-structure de canal> ::=
 <référence graphique de sous-structure de canal>
 | <diagramme de sous-structure de canal>

<référence graphique de sous-structure de canal> ::=
 <symbole de sous-structure de canal> **contains** <nom de sous-structure de canal>

<symbole de sous-structure de canal> ::=
 <symbole de bloc>

<diagramme de sous-structure de canal> ::=
 <symbole de cadre>
 contains { <en-tête de sous-structure de canal>
 { { <zone texte de sous-structure de canal> }*
 { <diagramme de macro> }*
 <zone interaction de blocs> }set }
 is associated with { <identificateur de bloc> | ENV }+

L'<identificateur de bloc> ou ENV identifie un point d'extrémité du canal subdivisé. L'<identificateur de bloc> est placé à l'extérieur du <symbole de cadre> près du point d'extrémité du sous-canal associé au <symbole de cadre>.

<en-tête de sous-structure de canal> ::=
 SUBSTRUCTURE { <nom de sous-structure de canal>
 | <identificateur de sous-structure de canal> }

<zone texte de sous-structure de canal> ::=
 <zone texte de système>

C1.9.1 Diagramme de macro

<diagramme de macro> ::=
 <symbole de cadre> **contains** <en-tête de macro> <zone de corps de macro>

<en-tête de macro> ::=
 MACRODEFINITION <nom de macro> [<paramètres formels de macro>]

<zone de corps de macro> ::=
 { { <zone quelconque> } *
 <zone quelconque> [**is connected to** <port de corps1 de macro>] **set**
 | { <zone quelconque> **is connected to** <port de corps2 de macro>
 <zone quelconque> **is connected to** <port de corps2 de macro>
 { <zone quelconque> [**is connected to** <port de corps2 de macro>] } * **set**

<symbole de port d'entrée de macro> ::=



<symbole de port de sortie de macro> ::=



<port de corps1 de macro> ::=
 <symbole de port de sortie> [**is associated with** <étiquette de macro>]
is connected to { <symbole de cadre>
 | <symbole de port d'entrée de macro>
 | <symbole de port de sortie de macro> }

<port de corps2 de macro> ::=
 <symbole de port de sortie> **is associated with** <étiquette de macro>
is connected to { <symbole de cadre>
 | <symbole de port d'entrée de macro>
 | <symbole de port de sortie de macro> }

<étiquette de macro> ::=
 <nom>

<symbole de sortie> ::=
 <symbole de port de sortie fictif>
 | <symbole de ligne de flot>
 | <symbole de canal>
 | <symbole d'acheminement de signal>
 | <symbole d'association continu>
 | <symbole d'association pointillé>
 | <symbole de ligne de création>

<symbole de port de sortie fictif> ::=
 <symbole d'association continu>

```

<zone quelconque> ::=
    <zone texte de système>
    <zone interaction de blocs>
    <zone de liste de signaux>
    <zone de bloc>
    <zone texte de bloc>
    <zone d'interaction de processus>
    <référence graphique de procédure>
    <zone de procédure>
    <zone texte de processus>
    <zone de graphe de processus>
    <zone de fusion>
    <zone de chaîne de transition>
    <zone d'état>
    <zone d'entrée>
    <zone de mise en réserve>
    <zone d'extension de texte>
    <zone association de sous-structure de canal>
    <zone de sous-structure de canal>
    <zone de sous-structure de bloc>
    <zone d'entrée prioritaire>
    <zone de signal continu>
    <zone de connecteur d'entrée>
    <zone état suivant>
    <zone de processus>
    <zone de définition de canal>
    <zone de ligne de création>
    <zone de définition d'acheminement de signal>
    <référence graphique de processus>
    <diagramme de processus>
    <zone de départ>
    <zone de sortie>
    <zone d'initialisation>
    <zone de réinitialisation>
    <zone d'export>
    <zone de sortie prioritaire>
    <zone de tâche>
    <zone de demande de création>
    <zone d'appel de procédure>
    <zone de décision>
    <zone de connecteur de sortie>
    <zone texte de procédure>
    <zone de graphe de procédure>
    <zone de départ de procédure>
    <zone texte de sous-structure de bloc>
    <zone interaction de blocs>
    <zone de service>
    <zone de définition d'acheminement de signal de service>
    <zone texte de service>
    <zone de graphe de service>
    <zone de départ de service>
    <zone de commentaire>
    <zone d'appel de macro>

```

C1.9.2 Appel de macro

```

<zone d'appel de macro> ::=
    <symbole d'appel de macro> contains { <nom de macro> [<corps d'appel de macro>] }
    [is connected to
    { <port d'appel1 de macro> | <port d'appel2 de macro> { <port d'appel2 de macro> }+ } ]

```

<symbole d'appel de macro > ::=



<port d'appel1 de macro > ::=
 <symbole de port d'entrée> [is associated with <étiquette de macro>]
 is connected to <zone quelconque>

<port d'appel2 de macro > ::=
 <symbole de port d'entrée> is associated with <étiquette de macro>
 is connected to <zone quelconque>

<symbole de port d'entrée > ::=
 <symbole de port d'entrée fictif>
 | <symbole de ligne de flot >
 | <symbole de canal >
 | <symbole d'acheminement de signal >
 | <symbole d'association continu >
 | <symbole d'association pointillé >
 | <symbole de ligne de création >


<symbole de port d'entrée fictif > ::=
 <symbole d'association continu >

C1.10 *Systèmes génériques*

C1.10.1 *Définition optionnelle*

<zone d'option > ::=
 <symbole d'option > contains
 { SELECT IF (<expression booléenne simple >)
 { <zone de bloc >
 | <zone de sous-structure de canal >
 | <zone texte de système >
 | <zone texte de bloc >
 | <zone texte de processus >
 | <zone texte de procédure >
 | <zone texte de sous-structure de bloc >
 | <zone texte de sous-structure de canal >
 | <zone texte de service >
 | <diagramme de macro >
 | <zone d'option >
 | <zone de processus >
 | <zone de définition d'acheminement de signal >
 | <zone de ligne de création >
 | <zone de procédure >
 | <zone de service >
 | <zone de définition d'acheminement de signal de service > }+ } }


Le <symbole d'option > est un polygone en pointillé dont les coins sont en trait plein, par exemple:



C1.10.1 Transition optionnelle

<zone d'option de transition> ::=
 <symbole d'option de transition> **contains** { <question alternative> }
 is followed by { <port de sortie1 d'option> { <port de sortie2 d'option> }
 { <port de sortie1 d'option> }* }set

<symbole d'option de transition> ::=



<port de sortie1 d'option> ::=
 <symbole de ligne de flot> **is associated with** <réponse graphique>
 is followed by { <zone de transition> | <zone de fusion> }

<port de sortie2 d'option> ::=
 <symbole de ligne de flot> **is associated with** ELSE
 is followed by { <zone de transition> | <zone de fusion> }

C1.11 Service

C1.11.1 Décomposition de service

<zone d'interaction de service> ::=
 { <zone de service> | <zone de définition d'acheminement de signal de service> }+

<zone de service> ::=
 <référence graphique de service>
 | <diagramme de service>

<référence graphique de service> ::=
 <symbole de service> **contains** <nom de service>

<symbole de service> ::=



<zone de définition d'acheminement de signal de service> ::=
 <symbole d'acheminement de signal>
 is associated with { <nom d'acheminement de signal de service>
 [<identificateur d'acheminement de signal>]
 <zone de liste de signaux> [<zone de liste de signaux>] }set
 is connected to { <zone de service>
 { <zone de service> | <symbole de cadre> } }set

Lorsque le <symbole d'acheminement de signal> est relié au <symbole de cadre>, l'<identificateur d'acheminement de signal> identifie alors un acheminement de signal externe auquel l'acheminement de signal est relié.

C1.11.2 Diagramme de service

<diagramme de service> ::=
 <symbole de cadre> **contains**
 { <en-tête de service>
 { { <zone texte de service> }*
 { <zone de procédure> }*
 { <diagramme de macro> }*
 <zone de graphe de service> }set }

<en-tête de service> ::=
 SERVICE { <nom de service> | <identificateur de service> }

<zone texte de service> ::=
 <symbole de texte> **contains**
 { <définition de variable>
 | <définition de vue>
 | <définition d'import>
 | <définition de données>
 | <définition de macro>
 | <définition de temporisateur>
 | <définition de sélection> }*

<zone de graphe de service> ::=
 <zone de graphe de processus>

<zone d'association d'entrée prioritaire> ::=
 <symbole d'association continu> **is connected to** <zone d'entrée prioritaire>


<zone d'entrée prioritaire> ::=
 <symbole d'entrée prioritaire> **contains** <liste d'entrée prioritaire>

<symbole d'entrée prioritaire> ::=



<zone de sortie prioritaire> ::=
 <symbole de sortie prioritaire> **contains** <corps de sortie prioritaire>

<symbole de sortie prioritaire> ::=



C1.12 *Signal continu*

<zone association de signal continu> ::=
 <symbole d'association continu> **is connected to** <zone de signal continu>

<zone de signal continu> ::=
 <symbole de condition de validation>
contains {<expression booléenne> [<fin> PRIORITY <nom de littéral entier>]}
is followed by <zone de transition>

C1.13 *Condition de validation*

<zone de condition de validation> ::=
 <symbole de condition de validation> **contains** <expression booléenne>

<symbole de condition de validation> ::=



C1.14 *Export*

<zone d'export> ::=
 <symbole de tâche> **contains** <export>

SÉRIES DES RECOMMANDATIONS UIT-T

Série A	Organisation du travail de l'UIT-T
Série B	Moyens d'expression: définitions, symboles, classification
Série C	Statistiques générales des télécommunications
Série D	Principes généraux de tarification
Série E	Exploitation générale du réseau, service téléphonique, exploitation des services et facteurs humains
Série F	Services de télécommunication non téléphoniques
Série G	Systèmes et supports de transmission, systèmes et réseaux numériques
Série H	Systèmes audiovisuels et multimédias
Série I	Réseau numérique à intégration de services
Série J	Transmission des signaux radiophoniques, télévisuels et autres signaux multimédias
Série K	Protection contre les perturbations
Série L	Construction, installation et protection des câbles et autres éléments des installations extérieures
Série M	RGT et maintenance des réseaux: systèmes de transmission, de télégraphie, de télécopie, circuits téléphoniques et circuits loués internationaux
Série N	Maintenance: circuits internationaux de transmission radiophonique et télévisuelle
Série O	Spécifications des appareils de mesure
Série P	Qualité de transmission téléphonique, installations téléphoniques et réseaux locaux
Série Q	Commutation et signalisation
Série R	Transmission télégraphique
Série S	Equipements terminaux de télégraphie
Série T	Terminaux des services télématiques
Série U	Commutation télégraphique
Série V	Communications de données sur le réseau téléphonique
Série X	Réseaux de données et communication entre systèmes ouverts
Série Y	Infrastructure mondiale de l'information et protocole Internet
Série Z	Langages et aspects informatiques généraux des systèmes de télécommunication